
Stata tip 113: Changing a variable’s format: What it

does and does not mean
Nicholas J. Cox
Department of Geography
Durham University
Durham, UK

n.j.cox@durham.ac.uk

1 Introduction

Stata variables are all associated with a display format. Users can assign such a display
format or work with a display format assigned by default. Thus suppose that you create
new variables, for example,

. set obs 1
obs was 0, now 1

. generate mynum = 42

. generate mystr = "42"

Now type

. describe

Contains data
obs: 1
vars: 2
size: 6

storage display value
variable name type format label variable label

mynum float %9.0g
mystr str2 %9s

Sorted by:
Note: dataset has changed since last saved

You can see that mynum has been created as a float variable with display format %9.0g
and that mystr has been created as a str2 variable with display format %9s. If you do
not like either of those formats, you can change them using the format command (see
[D] format).

The problem addressed in this tip is that users are often puzzled over exactly what
is meant by changing formats. Part of the problem behind such puzzlement may be
linguistic. Sometimes, the term “format” is used vaguely or loosely, such as when
“formatting” implies something like initial preparation or transformation of the data.
At other times, the term may be used precisely but not in the sense of Stata’s format

command. Thus you may read of long or wide format in the sense of dataset shape or

c© 2012 StataCorp LP dm0067



762 Stata tip 113

structure. Part of the solution to such puzzlement is thus also linguistic, to remember
that “format” here means “display format”.

2 Applying format

Consider auto.dta, in particular the gear ratio variable.

. sysuse auto, clear
(1978 Automobile Data)

. codebook gear_ratio

gear_ratio Gear Ratio

type: numeric (float)

range: [2.19,3.89] units: .01
unique values: 36 missing .: 0/74

mean: 3.01486
std. dev: .456287

percentiles: 10% 25% 50% 75% 90%
2.43 2.73 2.955 3.37 3.72

codebook usefully reports that this variable has units (some say “resolution”: for
example, Murphy [1997]) of 0.01, meaning that values for this variable are given to 2
decimal places. Regardless of that, from the results of

. summarize gear_ratio

Variable Obs Mean Std. Dev. Min Max

gear_ratio 74 3.014865 .4562871 2.19 3.89

we can see that the default display of summarize shows many decimal places. The mean
is reported to 6 decimal places and the standard deviation to 7, which is many more
than are present in the original data. It can be asserted confidently that summarize

shows far more minute detail than anyone can use or interpret. However, typing

. describe gear_ratio

storage display value
variable name type format label variable label

gear_ratio float %6.2f Gear Ratio

shows that gear ratio has a display format of %6.2f. If we wish summarize to honor
that display format, we must specify the format option:

. summarize gear_ratio, format

Variable Obs Mean Std. Dev. Min Max

gear_ratio 74 3.01 0.46 2.19 3.89

Now the results may be a little too Spartan for some tastes. One common suggestion
is that a standard deviation can be reported a little more precisely than the original
data. To get a more precise display, we can change the format of gear ratio:

. format gear_ratio %6.3f

We see the result when we reissue the summarize command:

. summarize gear_ratio, format

Variable Obs Mean Std. Dev. Min Max

gear_ratio 74 3.015 0.456 2.190 3.890

This example is telling. The first thing to do is look for an option that changes the
format of results. If there is not one, or it does not do what you want, then use format

directly and reissue the command.

Moreover, it should be clear that format cannot introduce a third decimal place to
the gear ratio data that was never typed in originally, that is, at the time auto.dta

was compiled. All format does is change how data and results are displayed. Even if
the format is changed to one coarser than was entered, a detailed check will confirm
that the data themselves are unchanged. One way to see this, left as an exercise for
you, is to vary the format of some key variable and then check that the results from
some interesting command remain completely identical. Use return list or ereturn

list to get a high-resolution display.

3 Date formats

Dates are common in many problems but often do not arrive in exactly the right form
for analysis. Hence users often want to convert dates as received to some other kind of
date.

A common misconception is that changing the date format is the way to change one
kind of date to another kind. That is wrong. Suppose we wish to change daily dates to
monthly dates. As an experiment, set up a daily date variable:

. clear

. set obs 1
obs was 0, now 1

. generate mydate = d(8oct2012)

. format mydate %td

. list

mydate

1. 08oct2012

Now change the format to monthly:

. format mydate %tm

. list

mydate

1. 3566m3

You may think that the data have changed but to something absurd, so how did
Stata mess up? In fact, the data have not changed. The data are still an integer value
of 19274, which is October 8, 2012, when you count days from a zero date of January 1,
1960. The integer value 19274 is also March 3566 when you count months from a zero
date of January 1960. The result is not what we may have wanted or expected, but
Stata’s point of view is that it gave us what we asked for, the same data value but
interpreted according to a different format.

To convert a date, we need to use an appropriate conversion function. Then, and
only then, will format work to produce nicer displays:

. generate mymonth = mofd(mydate)

. format mymonth %tm

. format mydate %td

. list

mydate mymonth

1. 08oct2012 2012m10

